In[637]:=
```
Remove["Global`*"]
```

II. Dynamics

Quadrotor

Attitude

The attitude is described by three Euler angles. We define general rotation matrices around the x--, y--, and z--axis:

In[638]:=
```
Rx[val_] := {{1, 0, 0}, {0, Cos[val], -Sin[val]}, {0, Sin[val], Cos[val]}}
Ry[val_] := {{Cos[val], 0, Sin[val]}, {0, 1, 0}, {-Sin[val], 0, Cos[val]}}
Rz[val_] := {{Cos[val], -Sin[val], 0}, {Sin[val], Cos[val], 0}, {0, 0, 1}}
```

Rotational motion

The Euler angle rates are measured in their respective coordinate systems. The vehicle's control inputs are the rates in the vehicle's coordinate system {V}. The transformation is done using the Euler angles :

In[641]:=
```
eulertobody = {{γ'[t]}, {0}, {0}} +
  Rx[γ[t]]ᵀ.{{0}, {β'[t]}, {0}} + (Ry[β[t]].Rx[γ[t]])ᵀ.{{0}, {0}, {α'[t]}}
```

Out[641]= {{-Sin[β[t]] α′[t] + γ′[t]}, {Cos[β[t]] Sin[γ[t]] α′[t] + Cos[γ[t]] β′[t]}, {Cos[β[t]] Cos[γ[t]] α′[t] - Sin[γ[t]] β′[t]}}

Translational motion

The vehicle motion is given by

In[642]:=
```
quadtrans = {{x''[t]}, {y''[t]}, {z''[t]}} ==
  Rz[α[t]].Ry[β[t]].Rx[γ[t]].{{0}, {0}, {a[t]}} + {{0}, {0}, {-g}}
```

Out[642]= {{x″[t]}, {y″[t]}, {z″[t]}} ==
{{a[t] (Cos[α[t]] Cos[γ[t]] Sin[β[t]] + Sin[α[t]] Sin[γ[t]])},
{a[t] (Cos[γ[t]] Sin[α[t]] Sin[β[t]] - Cos[α[t]] Sin[γ[t]])},
{-g + a[t] Cos[β[t]] Cos[γ[t]]}}

Pendulum

Lagrangian

The Lagrangian is the difference of the kinetic energy and the potential energy. To show independency of the pendulum mass, we include is (named m_p) in this derivation :

In[643]:=
```
ζ[t] = Sqrt[L^2 - r[t]^2 - s[t]^2]
ℒ[t] = 1/2 * mp * ((x'[t] + r'[t])^2 + (y'[t] + s'[t])^2 + (z'[t] + D[ζ[t], t])^2) -
  mp * g * (z[t] + ζ[t])
```

Out[643]= $\sqrt{L^2 - r[t]^2 - s[t]^2}$

Out[644]= $-g\, m_p \left(\sqrt{L^2 - r[t]^2 - s[t]^2} + z[t]\right) + \frac{1}{2} m_p \left((r'[t] + x'[t])^2 + (s'[t] + y'[t])^2 + \left(\frac{-2\, r[t]\, r'[t] - 2\, s[t]\, s'[t]}{2\sqrt{L^2 - r[t]^2 - s[t]^2}} + z'[t]\right)^2\right)$

We take the derivatives as shown in the paper and solve for the accelerations r''[t] and s''[t]:

In[645]:=
```
pendulumr = FullSimplify[Solve[D[D[ℒ[t], r'[t]], t] - D[ℒ[t], r[t]] == 0, r''[t]]]
pendulums = FullSimplify[Solve[D[D[ℒ[t], s'[t]], t] - D[ℒ[t], s[t]] == 0, s''[t]]]
```

Out[645]=
$$\Big\{\Big\{r''[t] \to \frac{1}{\left(L^2 - s[t]^2\right)\left(L^2 - r[t]^2 - s[t]^2\right)} \Big(-r[t]^4\, x''[t] - \left(L^2 - s[t]^2\right)^2 x''[t] - 2\, r[t]^2 \left(s[t]\, r'[t]\, s'[t] + \left(-L^2 + s[t]^2\right) x''[t]\right) + r[t]^3 \left(s'[t]^2 + s[t]\, s''[t] - \sqrt{L^2 - r[t]^2 - s[t]^2}\, (g + z''[t])\right) + r[t] \Big(-L^2\, s[t]\, s''[t] + s[t]^3\, s''[t] + s[t]^2 \left(r'[t]^2 - \sqrt{L^2 - r[t]^2 - s[t]^2}\, (g + z''[t])\right) + L^2 \left(-r'[t]^2 - s'[t]^2 + \sqrt{L^2 - r[t]^2 - s[t]^2}\, (g + z''[t])\right)\Big)\Big)\Big\}\Big\}$$

Out[646]=
$$\Big\{\Big\{s''[t] \to \frac{1}{\left(L^2 - r[t]^2\right)\left(L^2 - r[t]^2 - s[t]^2\right)} \Big(-\left(L^2 - r[t]^2\right)^2 y''[t] - s[t]^4\, y''[t] - 2\, s[t]^2 \left(r[t]\, r'[t]\, s'[t] + \left(-L^2 + r[t]^2\right) y''[t]\right) + s[t]^3 \left(r'[t]^2 + r[t]\, r''[t] - \sqrt{L^2 - r[t]^2 - s[t]^2}\, (g + z''[t])\right) + s[t] \Big(-L^2\, r[t]\, r''[t] + r[t]^3\, r''[t] + r[t]^2 \left(s'[t]^2 - \sqrt{L^2 - r[t]^2 - s[t]^2}\, (g + z''[t])\right) + L^2 \left(-r'[t]^2 - s'[t]^2 + \sqrt{L^2 - r[t]^2 - s[t]^2}\, (g + z''[t])\right)\Big)\Big)\Big\}\Big\}$$

we see that the dynamics on the pendulum depend on r[t], s[t], r'[t], s'[t], and the vehicle accelerations x''[t], y''[t], z''[t]. We also see that the equations of motion are independent of the pendulum mass.

III. Nominal trajectories

Constant position

In[647]:=
```
Solve[Rz[α[t]].Ry[β[t]].Rx[γ[t]].{{0}, {0}, {a[t]}} + {{0}, {0}, {-g}} == {{0}, {0}, {0}},
  {β[t], γ[t], a[t]}]
```

Solve::ifun :
Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

Out[647]=
```
{{a[t] → -g, γ[t] → 0, β[t] → -π},
 {a[t] → -g, γ[t] → 0, β[t] → π}, {a[t] → -g, γ[t] → -π, β[t] → 0},
 {a[t] → -g, γ[t] → π, β[t] → 0}, {a[t] → g, γ[t] → 0, β[t] → 0},
 {a[t] → g, γ[t] → -π, β[t] → -π}, {a[t] → g, γ[t] → -π, β[t] → π},
 {a[t] → g, γ[t] → π, β[t] → -π}, {a[t] → g, γ[t] → π, β[t] → π}}
```

In[648]:=
```
nominalconditionsstanstill = Solve[FullSimplify[
    {D[D[ℒ[t], r'[t]], t] - D[ℒ[t], r[t]] == 0, D[D[ℒ[t], s'[t]], t] - D[ℒ[t], s[t]] == 0},
    {x''[t] == 0, y''[t] == 0, z''[t] == 0, r''[t] == 0,
     r'[t] == 0, s''[t] == 0, s'[t] == 0}], {r[t], s[t]}]
```

Out[648]=
```
{{r[t] → 0, s[t] → 0}}
```

Circular trajectory

We introduce the coordinate transformations on translational quadrotor position, translational quadrotor position, and quadrotor attitude, as described in the paper:

In[649]:=
```
transattquad =
  Rz[α[t]].Ry[β[t]].Rx[γ[t]].{{0}, {0}, {1}} == Rz[η[t]].Ry[μ[t]].Rx[ν[t]].{{0}, {0}, {1}}
transocquad = {{x[t]}, {y[t]}, {z[t]}} == Rz[Ω * t].{{u[t]}, {v[t]}, {w[t]}}
transocstick = {{r[t]}, {s[t]}, {0}} == Rz[Ω * t].{{p[t]}, {q[t]}, {0}}
```

Out[649]=
```
{{Cos[α[t]] Cos[γ[t]] Sin[β[t]] + Sin[α[t]] Sin[γ[t]]},
  {Cos[γ[t]] Sin[α[t]] Sin[β[t]] - Cos[α[t]] Sin[γ[t]]}, {Cos[β[t]] Cos[γ[t]]}} ==
 {{Cos[η[t]] Cos[ν[t]] Sin[μ[t]] + Sin[η[t]] Sin[ν[t]]},
  {Cos[ν[t]] Sin[η[t]] Sin[μ[t]] - Cos[η[t]] Sin[ν[t]]}, {Cos[μ[t]] Cos[ν[t]]}}
```

Out[650]=
```
{{x[t]}, {y[t]}, {z[t]}} ==
 {{Cos[t Ω] u[t] - Sin[t Ω] v[t]}, {Sin[t Ω] u[t] + Cos[t Ω] v[t]}, {w[t]}}
```

Out[651]=
```
{{r[t]}, {s[t]}, {0}} == {{Cos[t Ω] p[t] - q[t] Sin[t Ω]}, {Cos[t Ω] q[t] + p[t] Sin[t Ω]}, {0}}
```

It is then possible to rewrite the quadrotor equation of motion in the rotating coordinate system:

In[652]:=
```
quadmotionC = Solve[Eliminate[
    {FullSimplify[quadtrans /. Flatten[{Solve[transocquad, {x[t], y[t], z[t]}],
          Solve[D[transocquad, t], {x'[t], y'[t], z'[t]}],
          Solve[D[D[transocquad, t], t], {x''[t], y''[t], z''[t]}] }]],
      Rz[α[t]].Ry[β[t]].Rx[γ[t]].{{0}, {0}, {a[t]}} ==
       Rz[Ω * t].Ry[μ[t]].Rx[ν[t]].{{0}, {0}, {a[t]}}} /.
     α[t] → 0, {β[t], γ[t]}], {u''[t], v''[t], w''[t]}]
```

Eliminate::ifun :

Inverse functions are being used by Eliminate, so some solutions may not be found; use Reduce for complete solution information. ≫

Out[652]=
```
{{w''[t] → -g + a[t] Cos[μ[t]] Cos[ν[t]], u''[t] → a[t] Cos[ν[t]] Sin[μ[t]] + Ω^2 u[t] + 2 Ω v'[t],
  v''[t] → -a[t] Sin[ν[t]] + Ω^2 v[t] - 2 Ω u'[t]}}
```

We solve for the nominal states of the quadrotor:

In[653]:=
```
nominalquadcircle = FullSimplify[
  Solve[Flatten[Flatten[{u''[t] == 0, v''[t] == 0, w''[t] == 0} /. {quadmotionC /.
       {u'[t] → 0, v'[t] → 0, w'[t] → 0, u[t] → R, v[t] → 0}}]], {a[t], μ[t], ν[t]}]]
```

Solve::ifun :

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

Out[653]=

$$\left\{\left\{a[t]\to-\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to0,\ \mu[t]\to-\text{ArcCos}\left[-\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},\right.$$
$$\left\{a[t]\to-\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to0,\ \mu[t]\to\text{ArcCos}\left[-\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to-\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to-\pi,\ \mu[t]\to-\text{ArcCos}\left[\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to-\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to-\pi,\ \mu[t]\to\text{ArcCos}\left[\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to-\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to\pi,\ \mu[t]\to-\text{ArcCos}\left[\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to-\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to\pi,\ \mu[t]\to\text{ArcCos}\left[\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to0,\ \mu[t]\to-\text{ArcCos}\left[\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to0,\ \mu[t]\to\text{ArcCos}\left[\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to-\pi,\ \mu[t]\to-\text{ArcCos}\left[-\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to-\pi,\ \mu[t]\to\text{ArcCos}\left[-\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left\{a[t]\to\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to\pi,\ \mu[t]\to-\text{ArcCos}\left[-\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\},$$
$$\left.\left\{a[t]\to\sqrt{g^2+R^2\Omega^4},\ \nu[t]\to\pi,\ \mu[t]\to\text{ArcCos}\left[-\frac{g}{\sqrt{g^2+R^2\Omega^4}}\right]\right\}\right\}$$

In[654]:=
```
(*notice that, in the paper, we have chosen to represent μ[t] through its tangent,
rather than the cosine described here. Conversion
 is possible through simple triangular identities.*)
```

and then solve for the control inputs:

In[655]:=
```
eulerder = D[FullSimplify[
   transattquad /. Flatten[{{α[t] → 0}, nominalquadcircle[[10]], {η[t] → Ω * t}}]], t]
```

Out[655]=

$$\{\{\text{Cos}[\beta[t]]\,\text{Cos}[\gamma[t]]\,\beta'[t]-\text{Sin}[\beta[t]]\,\text{Sin}[\gamma[t]]\,\gamma'[t]\},$$
$$\{-\text{Cos}[\gamma[t]]\,\gamma'[t]\},\ \{-\text{Cos}[\gamma[t]]\,\text{Sin}[\beta[t]]\,\beta'[t]-\text{Cos}[\beta[t]]\,\text{Sin}[\gamma[t]]\,\gamma'[t]\}\}==$$
$$\left\{\left\{\Omega\sqrt{\frac{R^2\Omega^4}{g^2+R^2\Omega^4}}\ \text{Sin}[t\,\Omega]\right\},\ \left\{-\Omega\sqrt{\frac{R^2\Omega^4}{g^2+R^2\Omega^4}}\ \text{Cos}[t\,\Omega]\right\},\ \{0\}\right\}$$

```
In[656]:= eulerrates = FullSimplify[Solve[{eulerder[[1, 1]] == eulerder[[2, 1]],
      eulerder[[1, 2]] == eulerder[[2, 2]]}, {β'[t], γ'[t]}]]
```

Out[656]=
$$\left\{\left\{\beta'[t] \to \Omega \sqrt{\frac{R^2 \Omega^4}{g^2 + R^2 \Omega^4}}\ \mathrm{Sec}[\gamma[t]]\ (\mathrm{Sec}[\beta[t]]\ \mathrm{Sin}[t\,\Omega] + \mathrm{Cos}[t\,\Omega]\ \mathrm{Tan}[\beta[t]]\ \mathrm{Tan}[\gamma[t]]),\right.\right.$$
$$\left.\left.\gamma'[t] \to \Omega \sqrt{\frac{R^2 \Omega^4}{g^2 + R^2 \Omega^4}}\ \mathrm{Cos}[t\,\Omega]\ \mathrm{Sec}[\gamma[t]]\right\}\right\}$$

```
In[657]:= FullSimplify[eulertobody /. Flatten[{eulerrates, α'[t] → 0}]]
```

Out[657]=
$$\left\{\left\{\Omega \sqrt{\frac{R^2 \Omega^4}{g^2 + R^2 \Omega^4}}\ \mathrm{Cos}[t\,\Omega]\ \mathrm{Sec}[\gamma[t]]\right\},\right.$$
$$\left\{\Omega \sqrt{\frac{R^2 \Omega^4}{g^2 + R^2 \Omega^4}}\ (\mathrm{Sec}[\beta[t]]\ \mathrm{Sin}[t\,\Omega] + \mathrm{Cos}[t\,\Omega]\ \mathrm{Tan}[\beta[t]]\ \mathrm{Tan}[\gamma[t]])\right\},$$
$$\left.\left\{-\Omega \sqrt{\frac{R^2 \Omega^4}{g^2 + R^2 \Omega^4}}\ \mathrm{Tan}[\gamma[t]]\ (\mathrm{Sec}[\beta[t]]\ \mathrm{Sin}[t\,\Omega] + \mathrm{Cos}[t\,\Omega]\ \mathrm{Tan}[\beta[t]]\ \mathrm{Tan}[\gamma[t]])\right\}\right\}$$

We also rewrite the equations of motion of the pendulum in the rotating coordinate system:

```
In[658]:= ℒc[t] = FullSimplify[ℒ[t] /. Flatten[{Solve[transocquad, {x[t], y[t], z[t]}],
       Solve[D[transocquad, t], {x'[t], y'[t], z'[t]}],
       Solve[transocstick, {r[t], s[t]}], Solve[D[transocstick, t], {r'[t], s'[t]}]}]]
```

Out[658]=
$$\frac{1}{2} m_p \left(-2 g \left(\sqrt{L^2 - p[t]^2 - q[t]^2} + w[t]\right) + \right.$$
$$(\mathrm{Sin}[t\,\Omega]\ (-\Omega\,(q[t] + v[t]) + p'[t] + u'[t]) + \mathrm{Cos}[t\,\Omega]\ (\Omega\,(p[t] + u[t]) + q'[t] + v'[t]))^2 +$$
$$(\mathrm{Cos}[t\,\Omega]\ (\Omega\,(q[t] + v[t]) - p'[t] - u'[t]) + \mathrm{Sin}[t\,\Omega]\ (\Omega\,(p[t] + u[t]) + q'[t] + v'[t]))^2 +$$
$$\left.\frac{\left(p[t]\ p'[t] + q[t]\ q'[t] - \sqrt{L^2 - p[t]^2 - q[t]^2}\ w'[t]\right)^2}{L^2 - p[t]^2 - q[t]^2}\right)$$

and get the equilibrium conditions in the rotating coordinate system:

```
In[659]:= FullSimplify[
   {D[D[ℒc[t], p'[t]], t] - D[ℒc[t], p[t]] == 0, D[D[ℒc[t], q'[t]], t] - D[ℒc[t], q[t]] == 0} /.
    {p'[t] → 0, q'[t] → 0, w''[t] → 0, u''[t] → 0, v''[t] → 0, q''[t] → 0,
     p''[t] → 0, u'[t] → 0, v'[t] → 0, u[t] → R, v[t] → 0}, mp > 0]
```

Out[659]=
$$\left\{R\,\Omega^2 + p[t]\left(\Omega^2 + \frac{g}{\sqrt{L^2 - p[t]^2 - q[t]^2}}\right) == 0,\ q[t]\left(\Omega^2 + \frac{g}{\sqrt{L^2 - p[t]^2 - q[t]^2}}\right) == 0\right\}$$

IV. Dynamics about nominal trajectories

In this section, we calculate the partial derivatives around the setpoint, to demonstrate the derivation of the equations of motion.

Constant position

In[660]:=
```
nomstateconst := {r[t] → 0, s[t] → 0, r'[t] → 0,
  s'[t] → 0, a[t] → g, α[t] → 0, γ[t] → 0, β[t] → 0, z''[t] → 0}
```

Quadrotor dynamics

In[661]:=
```
(*Partial derivatives of quadrotor translation *)
(*"motion" contains {{x''[t]}, {y''[t]}, {z''[t]}}*)
motion := Rz[α[t]].Ry[β[t]].Rx[γ[t]].{{0}, {0}, {a[t]}} + {{0}, {0}, {-g}}
{x̃''[t], ỹ''[t], z̃''[t]} =
 Flatten[{D[motion, a[t]] /. nomstateconst} * ã[t] +
   {D[motion, α[t]] /. nomstateconst} * α̃[t] +
   {D[motion, β[t]] /. nomstateconst} * β̃[t] +
   {D[motion, γ[t]] /. nomstateconst} * γ̃[t] ]
```

Out[662]= $\{g\,\tilde{\beta}[t],\ -g\,\tilde{\gamma}[t],\ \tilde{a}[t]\}$

In[663]:=
```
(*Partial derivatives of quadrotor rotation*)
{ω̃x[t], ω̃y[t], ω̃z[t]} = Flatten[
  {D[eulertobody, α'[t]] /. nomstateconst} * α̃'[t] +
   {D[eulertobody, β'[t]] /. nomstateconst} * β̃'[t] +
   {D[eulertobody, γ'[t]] /. nomstateconst} * γ̃'[t]]
```

Out[663]= $\{\tilde{\gamma}'[t],\ \tilde{\beta}'[t],\ \tilde{\alpha}'[t]\}$

In[664]:= Pendulum dynamics

Out[664]= dynamics Pendulum

In[665]:=
```
r̃''[t] = FullSimplify[D[r''[t] /. pendulumr, r[t]] /. nomstateconst] * r̃[t] +
  FullSimplify[D[r''[t] /. pendulumr, r'[t]] /. nomstateconst] * r̃'[t] +
  FullSimplify[D[r''[t] /. pendulumr, s[t]] /. nomstateconst] * s̃[t] +
  FullSimplify[D[r''[t] /. pendulumr, s'[t]] /. nomstateconst] * s̃'[t] +
  FullSimplify[D[r''[t] /. pendulumr, x''[t]] /. nomstateconst] * x̃''[t] +
  FullSimplify[D[r''[t] /. pendulumr, y''[t]] /. nomstateconst] * ỹ''[t] +
  FullSimplify[D[r''[t] /. pendulumr, z''[t]] /. nomstateconst] * z̃''[t]
```

Out[665]= $\left\{\frac{g\,\tilde{r}[t]}{\sqrt{L^2}} - g\,\tilde{\beta}[t]\right\}$

In[666]:=
```
s̃''[t] = FullSimplify[D[s''[t] /. pendulums, r[t]] /. nomstateconst] * r̃[t] +
  FullSimplify[D[s''[t] /. pendulums, r'[t]] /. nomstateconst] * r̃'[t] +
  FullSimplify[D[s''[t] /. pendulums, s[t]] /. nomstateconst] * s̃[t] +
  FullSimplify[D[s''[t] /. pendulums, s'[t]] /. nomstateconst] * s̃'[t] +
  FullSimplify[D[s''[t] /. pendulums, x''[t]] /. nomstateconst] * x̃''[t] +
  FullSimplify[D[s''[t] /. pendulums, y''[t]] /. nomstateconst] * ỹ''[t] +
  FullSimplify[D[s''[t] /. pendulums, z''[t]] /. nomstateconst] * z̃''[t]
```

Out[666]= $\left\{\frac{g\,\tilde{s}[t]}{\sqrt{L^2}} + g\,\tilde{\gamma}[t]\right\}$

Circular trajectory

In[667]:=
```
nomstatecircle := {p[t] → p₀, q[t] → 0, p'[t] → 0, q'[t] → 0, p''[t] → 0, q''[t] → 0,
  u[t] → -p₀ * (Ω^2 + g / Sqrt[L^2 - p₀^2]) / Ω^2, v[t] → 0, u'[t] → 0, v'[t] → 0,
  w'[t] → 0, u''[t] → 0, v''[t] → 0, w''[t] → 0, a[t] → a₀, ν[t] → 0, μ[t] → μ₀}
```

Quadrotor dynamics

In[668]:=
```
ũ''[t] =
 FullSimplify[D[u''[t] /. quadmotionC, u[t]] /. nomstatecircle] * ũ[t] +
  FullSimplify[D[u''[t] /. quadmotionC, u'[t]] /. nomstatecircle] * ũ'[t] +
  FullSimplify[D[u''[t] /. quadmotionC, v[t]] /. nomstatecircle] * ṽ[t] +
  FullSimplify[D[u''[t] /. quadmotionC, v'[t]] /. nomstatecircle] * ṽ'[t] +
  FullSimplify[D[u''[t] /. quadmotionC, w[t]] /. nomstatecircle] * w̃[t] +
  FullSimplify[D[u''[t] /. quadmotionC, w'[t]] /. nomstatecircle] * w̃'[t] +
  FullSimplify[D[u''[t] /. quadmotionC, a[t]] /. nomstatecircle] * ã[t] +
  FullSimplify[D[u''[t] /. quadmotionC, μ[t]] /. nomstatecircle] * μ̃[t] +
  FullSimplify[D[u''[t] /. quadmotionC, ν[t]] /. nomstatecircle] * ν̃[t]
```

Out[668]= $\{\mathrm{Sin}[\mu_0]\,\tilde{a}[t] + \Omega^2\,\tilde{u}[t] + \mathrm{Cos}[\mu_0]\,a_0\,\tilde{\mu}[t] + 2\,\Omega\,\tilde{v}'[t]\}$

In[669]:=
```
ṽ'' [t] =
 FullSimplify[D[v''[t] /. quadmotionC, u[t]] /. nomstatecircle] * ũ[t] +
  FullSimplify[D[v''[t] /. quadmotionC, u'[t]] /. nomstatecircle] * ũ'[t] +
  FullSimplify[D[v''[t] /. quadmotionC, v[t]] /. nomstatecircle] * ṽ[t] +
  FullSimplify[D[v''[t] /. quadmotionC, v'[t]] /. nomstatecircle] * ṽ'[t] +
  FullSimplify[D[v''[t] /. quadmotionC, w[t]] /. nomstatecircle] * w̃[t] +
  FullSimplify[D[v''[t] /. quadmotionC, w'[t]] /. nomstatecircle] * w̃'[t] +
  FullSimplify[D[v''[t] /. quadmotionC, a[t]] /. nomstatecircle] * ã[t] +
  FullSimplify[D[v''[t] /. quadmotionC, μ[t]] /. nomstatecircle] * μ̃[t] +
  FullSimplify[D[v''[t] /. quadmotionC, ν[t]] /. nomstatecircle] * ν̃[t]
```

Out[669]= $\{\Omega^2\,\tilde{v}[t] - a_0\,\tilde{\nu}[t] - 2\,\Omega\,\tilde{u}'[t]\}$

In[670]:=
```
w̃''[t] =
 FullSimplify[D[w''[t] /. quadmotionC, u[t]] /. nomstatecircle] * ũ[t] +
  FullSimplify[D[w''[t] /. quadmotionC, u'[t]] /. nomstatecircle] * ũ'[t] +
  FullSimplify[D[w''[t] /. quadmotionC, v[t]] /. nomstatecircle] * ṽ[t] +
  FullSimplify[D[w''[t] /. quadmotionC, v'[t]] /. nomstatecircle] * ṽ'[t] +
  FullSimplify[D[w''[t] /. quadmotionC, w[t]] /. nomstatecircle] * w̃[t] +
  FullSimplify[D[w''[t] /. quadmotionC, w'[t]] /. nomstatecircle] * w̃'[t] +
  FullSimplify[D[w''[t] /. quadmotionC, a[t]] /. nomstatecircle] * ã[t] +
  FullSimplify[D[w''[t] /. quadmotionC, μ[t]] /. nomstatecircle] * μ̃[t] +
  FullSimplify[D[w''[t] /. quadmotionC, ν[t]] /. nomstatecircle] * ν̃[t]
```

Out[670]= $\{\mathrm{Cos}[\mu_0]\,\tilde{a}[t] - \mathrm{Sin}[\mu_0]\,a_0\,\tilde{\mu}[t]\}$

Pendulum dynamics

In[671]:=
```
pendulump := Solve[D[D[ℒc[t], p'[t]], t] - D[ℒc[t], p[t]] == 0, p''[t]]
pendulumq := Solve[D[D[ℒc[t], q'[t]], t] - D[ℒc[t], q[t]] == 0, q''[t]]
```

In[673]:=
```
p̃''[t] = FullSimplify[
  FullSimplify[D[p''[t] /. pendulump, p[t]] /. nomstatecircle] * p̃[t] +
   FullSimplify[D[p''[t] /. pendulump, p'[t]] /. nomstatecircle] * p̃'[t] +
   FullSimplify[D[p''[t] /. pendulump, q[t]] /. nomstatecircle] * q̃[t] +
   FullSimplify[D[p''[t] /. pendulump, q'[t]] /. nomstatecircle] * q̃'[t] +
   FullSimplify[D[p''[t] /. pendulump, u[t]] /. nomstatecircle] * ũ[t] +
   FullSimplify[D[p''[t] /. pendulump, v[t]] /. nomstatecircle] * ṽ[t] +
   FullSimplify[D[p''[t] /. pendulump, w[t]] /. nomstatecircle] * w̃[t] +
   FullSimplify[D[p''[t] /. pendulump, u'[t]] /. nomstatecircle] * ũ'[t] +
   FullSimplify[D[p''[t] /. pendulump, v'[t]] /. nomstatecircle] * ṽ'[t] +
   FullSimplify[D[p''[t] /. pendulump, w'[t]] /. nomstatecircle] * w̃'[t] +
   FullSimplify[D[p''[t] /. pendulump, u''[t]] /. nomstatecircle] * ũ''[t] +
   FullSimplify[D[p''[t] /. pendulump, v''[t]] /. nomstatecircle] * ṽ''[t] +
   FullSimplify[D[p''[t] /. pendulump, w''[t]] /. nomstatecircle] * w̃''[t]]
(* this can be put in the form shown in the paper using 1-p0^2/L^2==ξ^2/L^2*)
```

Out[673]=

$$\left\{\frac{\left(\mathrm{Cos}[\mu_0]\, p_0 \sqrt{L^2 - p_0^2} + \mathrm{Sin}[\mu_0]\left(-L^2 + p_0^2\right)\right)\tilde{a}[t]}{L^2} + \frac{g\,\tilde{p}[t]}{\sqrt{L^2 - p_0^2}} - \frac{\mathrm{Sin}[\mu_0]\, a_0\, p_0 \sqrt{L^2 - p_0^2}\,\tilde{\mu}[t]}{L^2} + \frac{\mathrm{Cos}[\mu_0]\, a_0 \left(-L^2 + p_0^2\right)\tilde{\mu}[t]}{L^2} + \frac{\Omega\left(L^2 - p_0^2\right)\left(\Omega\,\tilde{p}[t] + 2\,\tilde{q}'[t]\right)}{L^2}\right\}$$

In[674]:=
```
q̃''[t] =
 FullSimplify[D[q''[t] /. pendulumq, p[t]] /. nomstatecircle] * p̃[t] +
  FullSimplify[D[q''[t] /. pendulumq, p'[t]] /. nomstatecircle] * p̃'[t] +
  FullSimplify[D[q''[t] /. pendulumq, q[t]] /. nomstatecircle] * q̃[t] +
  FullSimplify[D[q''[t] /. pendulumq, q'[t]] /. nomstatecircle] * q̃'[t] +
  FullSimplify[D[q''[t] /. pendulumq, u[t]] /. nomstatecircle] * ũ[t] +
  FullSimplify[D[q''[t] /. pendulumq, v[t]] /. nomstatecircle] * ṽ[t] +
  FullSimplify[D[q''[t] /. pendulumq, w[t]] /. nomstatecircle] * w̃[t] +
  FullSimplify[D[q''[t] /. pendulumq, u'[t]] /. nomstatecircle] * ũ'[t] +
  FullSimplify[D[q''[t] /. pendulumq, v'[t]] /. nomstatecircle] * ṽ'[t] +
  FullSimplify[D[q''[t] /. pendulumq, w'[t]] /. nomstatecircle] * w̃'[t] +
  FullSimplify[D[q''[t] /. pendulumq, u''[t]] /. nomstatecircle] * ũ''[t] +
  FullSimplify[D[q''[t] /. pendulumq, v''[t]] /. nomstatecircle] * ṽ''[t] +
  FullSimplify[D[q''[t] /. pendulumq, w''[t]] /. nomstatecircle] * w̃''[t]
```

Out[674]=

$$\left\{\left(\Omega^2 + \frac{g}{\sqrt{L^2 - p_0^2}}\right)\tilde{q}[t] + a_0\,\tilde{\nu}[t] - 2\,\Omega\,\tilde{p}'[t]\right\}$$